
Урок 35. Модуль GraphABC. Пиксели и работа с цветом 

Пиксель (pixel) — наименьший элемент поверхности визуализации (например, экрана), которому 
может быть независимым образом заданы цвет, интенсивность и другие характеристики 
изображения. 

Пиксел (pixel — picture element) — черно-белые или цветные точки, на которые разделен экран 
монитора. Благодаря им, управляя их яркостью свечения, можно рисовать, чертить, строить 
графики. 

Пиксел/пиксель — неделимая точка в графическом изображении; наименьший адресуемый 
элемент растрового изображения. 

1. Подпрограммы для работы с пикселами 
 
procedure SetPixel(x,y: integer; c: Color); 

 Закрашивает пиксел с координатами (x, y) цветом c. 

procedure PutPixel(x,y: integer; c: Color); 

 Закрашивает пиксел с координатами (x, y) цветом c. 

function GetPixel(x,y: integer): Color; 

 Возвращает цвет пиксела с координатами (x, y). 

 

2. Подпрограммы для работы с цветом 
 
function RGB(r,g,b: byte): Color; 

 Возвращает цвет, который содержит красную (r), зеленую (g) и синюю (b) составляющие 
(r, g и b - в диапазоне от 0 до 255). 

function ARGB(a,r,g,b: byte): Color; 

 Возвращает цвет, который содержит красную (r), зеленую (g) и синюю (b) составляющие и 
прозрачность (a) (a, r, g, b - в диапазоне от 0 до 255). 

function clRandom: Color; 

 Возвращает случайный цвет. 

 
function RedColor(r: byte): Color; 

 Возвращает красный цвет с интенсивностью r (r - в диапазоне от 0 до 255). 

function GreenColor(g: byte): Color; 

 Возвращает зеленый цвет с интенсивностью g (g - в диапазоне от 0 до 255). 

function BlueColor(b: byte): Color; 

 Возвращает синий цвет с интенсивностью b (b - в диапазоне от 0 до 255). 

  



function GetRed(c: Color): integer; 

 Возвращает красную составляющую цвета. 

function GetGreen(c: Color): integer; 

 Возвращает зеленую составляющую цвета. 

function GetBlue(c: Color): integer; 

 Возвращает синюю составляющую цвета. 

function GetAlpha(c: Color): integer; 

 Возвращает составляющую прозрачности цвета. 

3. Пример программы (pixels.pas) 
program pixels; 
uses GraphABC; 
 
// Константы 
const 
  width = 600;    // Ширина окна 
  height = 600;   // Высота окна 
  red_x = 200;    // Центр красного круга: x 
  red_y = 213;    // Центр красного круга: y 
  green_x = 400;  // Центр зелёного круга: x 
  green_y = 213;  // Центр зелёного круга: y 
  blue_x = 300;   // Центр синего круга: x 
  blue_y = 386;   // Центр синего круга: y 
  radius = 180;   // Радиус кругов 
 
// Функция для вычисления расстояния между двумя точками 
function distance(x0, y0, x1, y1 : integer) : real; 
begin 
  // Sqrt вычисляет квадратный корень 
  Result := Sqrt((x0 - x1)**2 + (y0 - y1)**2); 
end; 
 
begin 
  SetWindowSize(width, height); 
  for var y := 0 to height - 1 do 
    for var x := 0 to width - 1 do 
    begin 
      var red : integer := 0; 
      var green : integer := 0; 
      var blue : integer := 0; 
      if distance(x, y, red_x, red_y) < radius then 
        red := 255; 
      if distance(x, y, green_x, green_y) < radius then 
        green := 255; 
      if distance(x, y, blue_x, blue_y) < radius then 
        blue := 255; 
      // Получить цвет из трёх составляющих 
      var c := RGB(red, green, blue); 
      // Закрасить пиксел 
      SetPixel(x, y, c); 
    end; 
end. 

